Decorate Your Web Pages with Mathematical Animations

Barbara K. D'Ambrosia
and
Carl R. Spitznagel

John Carroll University



Share some of the mathematics you love with the readers of your web pages! Maple V makes it easy to create mathematical animations, and those animations can be saved as animated GIF files that will make your web pages come to life! The image above is taken from an animation that rotates the graph of the function $f(x, y) = \sin(\sqrt{x^2 + y^2})$ in 3-space. Of course, an animation doesn't show up very well on paper—you will have to visit our web page to see this and other animations in action. Or—better yet—create some of your own using the techniques below.

Summary of Techniques:

1. ***Create an animation with Maple V.*** For example, the Maple code used to create the animated version of the image above is:

```
with(plots):
 pic := n ->
      [plot3d(sin(sqrt(x^2+y^2)),x=-5..5,y=-5..5,
       orientation=[3*n,67],grid=[20,20])]:
 display(seq(pic(n),n=0..29),insequence=true);
```

2. ***While interacting with Maple, reduce the size of the image.*** Grab one of the corner handles and drag. Remember that your image has multiple frames, so a large image will result in a very large file size—especially with 3D graphs.

3. ***While you're interacting with Maple, adjust the color of your animation.*** For a rotating 3D image, choose a color scheme in which the color of a point depends only on its z-coordinate. (We like to use Z(Hue).) This is especially important if you are taking advantage of symmetry in your image to make a 90 or 180 degree rotation seem like 360 degrees.

4. ***After you are satisfied with the appearance of your animation, choose File ... Export as ... HTML*** from the Maple menus. This process may take up to a minute or more, depending on how complex your animation is, and depending on the speed of your computer. Maple will save an HTML page that contains your Maple code and the image, stored as an animated GIF file. The Windows version of Maple saves the

GIF in a subfolder titled Images. Look there for the image when you get ready to include it in a web page. Maple may actually generate several GIF files, all with similar names like “rotate1.gif,” “rotate2.gif,” etc. One of these files is the animation; the others are graphic images of any mathematical equations in the output from your Maple commands.

5. ***Use an animation editor to do any additional editing to your image.*** Some of the touch-up editing that you may wish to do to your image:
   - *Slow it down!* With today's computers getting faster and faster, your animation may be moving at warp speed on some viewers’ machines.
   - *Add transparency to the GIF.* This is important if you like to write web pages with backgrounds other than plain white.
   - *Make it pause at the end*, before beginning the animation cycle again. For example, in our trefoil knot animation we pause after the knot is completely tied, before tying it all over again.

   There are a number of animation editors on the market. We like to use Animation Shop, which comes with Paint Shop Pro. (See http://www.jasc.com.)

6. ***Include your animation on your web page.*** All of the commonly used HTML editors include a way for you to insert images in web pages. Since animations created in this way are simply animated GIF files, no special handling is needed to insert the image.

7. ***Enjoy your new web page!!***

Examples of Animations:

In each of the examples on the following pages, the Maple V code can be copied and pasted into your own copy of Maple, if you would like to reproduce or experiment with the images. Live versions can be viewed on our web pages.

Conclusion:

Inclusion of mathematical animations on your web pages can help bring life to otherwise static pages. However, we caution you not to over-use these techniques, as we firmly believe that the effectiveness of these animations is greatly reduced by using them too frequently. Too many spinning, flapping or twisting images can be distracting to the readers of your web pages.

The authors can be contacted by e-mail at bdambrosia@jcu.edu and spitz@jcu.edu. Live examples of the animations discussed in this paper can be found at http://www.jcu.edu/math/ictcm99/.

Rotating 3D Surface

<table>
<tr>
<td>

Maple V code:

```
with(plots):
pic := n ->
    [plot3d(sin(sqrt(x^2+y^2)),x=-5..5,y=-5..5,
     orientation=[3*n,67],grid=[20,20])]:
display(seq(pic(n),n=0..29),insequence=true);
```

</td>
<td></td>
</tr>
<tr>
<td colspan="2">

Comments:

- It is actually the *viewpoint* that is being rotated by this Maple code, rather than the surface. However, the effect is that of the surface rotating in 3-space.
- The 4-way symmetry of this surface makes it possible to get by with rotating through only 90 degrees, rather than 360. This results in a considerably smaller animated GIF file. If you use a surface with less symmetry, you will have to rotate through a larger angle, resulting in more frames in the animation and a larger file size.
- Choose a color scheme in which the color of a point is determined by its z-coordinate; otherwise, there may be an abrupt color change when the rotation passes 90 degrees and starts again at 0 degrees. The color may be selected after Maple draws the graph. (We like to use Z(Hue).)

</td>
</tr>
</table>

Parametrized 3D Surface

<table>
<tr>
<td>

Maple V code:

```
with(plots):
animate3d((1-2*abs(1-t))*x*exp(-x^2-y^2),x=-3..3,
           y=-3..3,t=0..2,frames=20);
```

</td>
<td></td>
</tr>
<tr>
<td colspan="2">

Comments:

- The peak and valley of the surface $z = xe^{-x^2-y^2}$ are extended by multiplying by a parametrized function that takes values from -1 to 1, and then back to -1. Maple's “animate3d” command does the rest!
- If this animation will be viewed by a fairly fast computer, you will need to use an animation editor to slow down the resulting GIF file; otherwise, the pumping action of the surface may be distractingly fast. Our animation of this surface displays each frame for 0.15 seconds.

</td>
</tr>
</table>

Trefoil Knot

Maple V code:

```
with(plots):
TieKnot := proc(n :: posint)
    local i, t, curve, picts;
    curve := [-10*cos(t) - 2*cos(5*t) + 15*sin(2*t),
          -15*cos(2*t) + 10*sin(t) - 2*sin(5*t),
           10*cos(3*t)]:
    picts := [seq(tubeplot(curve,t=0..2*Pi*i/n,
          radius=3), i=1..n)];
    display(picts, insequence=true);
end;
TieKnot(48);
```



Comments:

- Maple's tubeplot command draws a "tube" around a parametric curve--in this case, the curve for a trefoil knot.
- The image above was created with 48 frames. You can experiment with other numbers of frames, as well as with various parametric curves and various values for the radius of the tube.
- An animation editor was used to slow down the display of the frames, as well as to add a 3-second display time to the final frame.
- This code was taken from pp. 317-318 of the Maple V Release 5 Programming Guide.

Riemann Sums Converging to an Integral

Maple V code:

```
with(plots): with(student):
 pic := n->
     [rightbox(x+sin(x)+3,x=0..2*Pi,2*n,
      tickmarks=[0,0])]:
 display(seq(pic(n),n=1..20),insequence=true);
```



Comments:

This animation shows Riemann sums for a function, using n = 2, 4, 6, 8, ..., 40 subintervals. It will have to be slowed down for display on almost all computers. Our version displays frames for varying times, the first and last frames being displayed for the longest times.

Solid of Revolution

Maple V code:	
see code below	

```
with(plots): with(plottools):
f := t -> t^3/2-2*t^2+2*t+1:
start:=spacecurve([0,t,f(t)],t=0..3,thickness=3):
pic := n ->
     cylinderplot(f(z),theta=0..n*2*Pi/30,z=0..3):
display(start,seq(rotate(pic(n),Pi/2,Pi/2,0),n=1..30),
     insequence=true,axes=normal,tickmarks=[0,0,0]);
```

Comments:

- This animation consists of 31 frames. The first is a spacecurve drawn in the y-z plane, and the next thirty frames are increasingly larger segments of the corresponding solid of revolution, drawn using Maple's "cylinderplot" routine.
- The viewpoint and color scheme were adjusted manually in Maple after the animation was constructed.
- We used an animation editor to slow down the animation, and to display the first and last frames for a longer time than the intermediate ones. This effect gives the viewer time to understand what is happening.
- Change the definition of the function in the second line, to experiment with other curves and their solids of revolution.

Contour Map/Surface

Maple V code:	
see code below	

```
with(plots):
pic := n ->
     [plot3d(x^2-y^2,x=-5..5,y=-5..5,
      orientation=[5*n/4,n], grid=[20,20],
      style=patchcontour)]:
display(seq(pic(n),n=0..60),insequence=true);
```

Comments:

- The familiar surface $z = x^2 - y^2$ is graphed with contours, while the viewpoint is varied from directly above on the z-axis to a location in the first octant. The effect is that of a 2-dimensional contour map changing into a surface in 3-space.
- In order to make the animation smooth, we use a large number (61) of frames, thereby making the file size fairly large.
- If this animation will be viewed by a fairly fast computer, you will need to use an animation editor to slow down the resulting GIF file. In our version of the animation, we display each frame for 0.1 seconds. (The first frame and the last frame are displayed for longer times, so that the viewer has time to understand what is happening.)